
Abstract

The use of random numbers in small microcontrollers can be very useful and fun in
robotic programming to yield different batiors from the same robotwontrol program.

Due to limited memory, the lack of floatingoint math capabilities, and the limited size

of integers in some microcontrollers, the practical implementation of a soffased

random number generator for a macontroller is difficult. This paper looks at the
generation of random numbers, and compares a variety of implementations on the
68HC11 microcontroller. A good implementation is chosen and the 68HC11 source code
is provided that implements it.

Introduction

The use of random numbers in robotics projects (and other microcontroller projects) can
be very beneficial. I've seen cases where a sM@lbwing robot gets stuck in a corner at
just the right antg. This is due to the algorithm hawbred into its brain producing the
exact same baeup, turna-bit, go-forward motion that just oscillates baakdforth in

the corner (what a disgrace!). While those of us who know actually what is going on
inside ofthe robot can understand this behavior, to the general public this makes the
robot appear to be quite brainless and diftted; the silly thing can’t even find its way
out of a corner. If the program in the robot used random numbers to slightly alter the
backup time and also the degrees turned in the same algorithm, this robot would not
have the exact same baakdforth behavior that caused it to be stuck in the corner. It
would very quickly maneuver out of the corner and continue on its way, making itsel
(and you) look quite intelligent. The availability of random numbers is also critical for
many algorithms incorporating artificial intelligence (Al) techniques such as genetic
algorithms and genetic programming, neural networks, and fuzzy logic.

The topc of random number generation (RNG) can be divided into two categories: true
random number generation and pseudo random number generation. With true random
number generation the next random number generated is not known, and the sequence of
random numbersannot be rggenerated. With pseudo random number generation, a
sequence of “random numbers” is generated using a known algorithm, and the exact same
sequence can be-generated; hence the classification of pseudo. For softhased

systems it is very @sirable to use a pseudo random number generator to be able to test

the software system with a repeatable set of random numbers; you-oam agest with



the exact same random numbers being used. Also, with a pseudo random number
generator that has be&rmally evaluated, you can be sure of the attributes of the
resulting sequence of random numbers.

Random number generators provided with workstatiass computers return a-R or

a 64bit integer number, or a floatingoint number based on that valun small
microcontrollers the desired value from a random number generator is generaillyitan 8
or 16-bit number. | will focus my discussion on aAdt random number generator.

In my random number generator study | first defined a set of gradingrierias attributes

| wanted in a RNG. | then wrote a Java program on a PC to test various RNG algorithms
against these criteria, and wrote different algorithms in this program to generate
sequences of random numbers to test. | then used the knowledgd framethe

successful algorithms and data to write 68H&pEcific assembly language to

implement the RNG. Lastly | ran the algorithm on the 68HC11 and verified that it
generated the same sequence seen in the Java code.

Hardware Methods

A true random numlyegenerator can be implemented in hardware using a witisd1]
source. IN2] and[3] a hardware circuit is detailed which is used to generate random
numbers using a paof transistors. Irj4] another hardwarbased random number
generator is discussed. While quite useful in many robotic applications, a hardased
random number generator is out of the scope of the focus of this paper.

Requirements of a “Good” Random Number Generator
Two famous quotes on random numbers help us to set the stage for our discussion:

“Anyone who considers arithmetical methods of producing random digits is,
of course, in a state of sin.John von Neumann (19a1957).

“The generation of random numbers is too important to be left to chance.”
Robert R. Coveyou, Oak Ridge National Laboratory in Tennessee (Found in
Ivars Peterson, 'The Jungles of Randomness' pp 178).

In the “art” of generating random numbers witisaftwarebased algorithm, you must
determine the attributes you want in a random number generator, and then test candidate
random number generators against these grading criteria. It is difficult to define a set of
attributes that are simple to define, Batestable, and complete enough to avoid

sequences of numbers that follow a simple pattern, such as a simple sequence (0, 1, 2, 3,
4,5...) or an incremenrby-N sequence (0, 21, 42, 63, 84..Tablel shows the list of
attributes | usedh testing candidate random number generators targeted for the 68HC11
microcontroller.



Table 1. Attributes for grading a RNG sequence.

Attribute

Parameters

Rational

Number range

Numbers from O to 255 are
generated.

An 8-bit processor typicglly
works with 8bit integers.

Number range
average

The average of this number range
127.5.

Befinition used in subsequent
attributes. (259)/2

Cycle length

The length of the pseudo random
sequence shall be at least 16,384
(2*) to 32,78 (2", but preferably
will be 65,536 (29).

The longer the RNG cycle is
the more varied the resulting
behavior is.

Number coverage

All numbers in the range will be
included the same number of time
in the cycle. For example, for a
cycle length of 65,53@ll numbers
0 through 255 will occur 256 times

Through the cycle | want all
snumbers to occur an equal
number of times so the
average of the cycle is the
.average of the range, and to
ensure all possible values can
be encountered.

Large window
average

The average of 128 sequential
values in all selections from the
cycle will vary from the average by
at most 20% (i.e. 102 to 153).

To make sure that any
sequence of this size will
cover a good range of value

\*2J

Large window
min/max

The average of 128 sequéit
values in at least 1 selection will
vary from the average by at least
10% above and by another
selection by at least 10% below
(i.e. 114 to 140).

To make sure the window
averages are not all the same
or too similar.

Medium window
average

The averagefdi32 sequential value
in all selections from the cycle will
vary from the average by at most
40% (i.e. 75t0 179).

sTo make sure that any
sequence of this size will
cover a good range of value

U)

Medium window
min/max

The average of 32 sequential valu
in at least 1 selection will vary
from the average by at least 20%
above and by another selection by
at least 20% below (i.e. 102 to
153).

€Bo make sure the window
averages are not all the same
or too similar.

Small window
average

The average of 8 sequentialues
in all selections from the cycle will
vary from the average by at most

To make sure that any
sequence of this size will
cover a good range of value

UJ

75% (i.e. 31 to 223).

" A 16-bit number can be generated using two successibiz alues. We could add an attribute to
consider the goodness of a sequence ebit&alues generated this way, but | decided against this.



Attribute Parameters Rational

Small window The average of 8 sequential valugsTo make sure the window
min/max in at least 1 selgmn will vary averages are not all the same
from the average by at least 35% | or too similar.
above and by another selection by
at least 35% below (i.e. 79 to 175).

Repeated values | A specific value may or may not bel want to discuss this, but do
repeated in theycle. not plan to specify a
requirement or test for this.

Repeated sequengeThere is not any sequence of 4 or| 3 or more repeated numbers
more numbers that is repeated in | may be useful/interesting, but

the cycle. | don’t want more than 3.
Valueto-value A sequence made from the To make sure there are no
delta differences in the sequence valuespatterns in the deltas.

will meet the same window average
and repeated criteria.

Operations allowed The operations used to implenten| These are the commonly
the random number generator are offered microcontroller

8-bit add, subtract, shift, and, or, | operations, and are found in
exor, and &bit to 16-bit multiply. the 68HC11.

Size of code The size of the code to implement Smaller is better with the
the algorithmis to be as small as | limited program space
possibleat most 100 bytes long. | available in microcontrollers

Required RAM The number of bytes of RAM Smaller is better with the
required to be dedicated to the limited RAM space available
RNG should be as few as possible in microcontrollers.

and at most 8 ytes.

Approaches

| investigated two different approaches to implementing a simple random number
generator for the 68HC11; a tabb@ased method and an algoritHimsed method.

Table-based method

For a table of 256 values, there are 256!, or 8.578e+506, possible ways to order the table.
My table-based approach uses a table of 128 values, with an algorithm to traverse the
table in a number of ways to generate a random number cy¢d84@umbers in length. |

used a table of 128 values and generated the other 128 values in the set28%pby
EXORIing the 128 table values with $FF. An acceptable set of 128 values for this table
include the criteria that EXORIing these 128 values gd@herate the remaining 128

values in the set. This gave me an effective table of 256 values using only 128 bytes, plus
a small number of bytes for the implementation logic. | then traverse this set of 256
values in 64 different ways to generate a sequefid®,384 values. Once the driving

logic of traversing the table was developed, a good set of table values as Jedata



was determined using a program to try different sets of valaies testing the resulting
table against thepecified grading criteria. | wrote a Java program to do this.

Table 2. 128 values used in my tableased RNG.

247, 86,108,103,252, 35,115, 75,202, 70,107, 89, 37, 40,246, 69

111,234, 56, 12,249,146, 19, 80,240,230, 45, 38,19 7,223, 65,123
26,208, 74,167,130, 79,253,121,173, 93,136,198, 64,204, 90, 20

30,233,222, 16, 14,242,182,174,195,105,159, 0, 84,129,250,155

10, 28,251, 92, 62,201,192,142,104,168,114, 49,172,124, 39, 67

143,254, 7,228,116, 31,154,156, 72, 3 6,194,161, 71,226, 59, 18
209,177,118,189,221, 98,135,122, 48,153,117,231,238,164, 52,187
44,145,170,213, 77, 97,128,212, 41,160, 11,149,200, 23, 50,179

As this was the first algorithm | got working, | was very attached to it. | then decided to
investigate equatichased methods for completeness in my research, but | really liked
my tablebased solution and thought | would decide to go with it as the best solution.

Equation -based method

An equationbased method uses a seed value. For each randobarngenerated this
seed value is used in an algorithm to generate a new seed value and also the random
number. In the case of ourldit random number generator we want the seed value to be
larger than the ®it number returned since anlft seed value Wi allow at most a cycle

of 256 values. A classic algorithm for generating random nunfibeis

unsigned long seed ;
seed = 16645251 * seed + 1013904223 ;

| implement code using a 46it seed value, and looked at diffete8-bit selections from
the newlycalculated seed as theb® random number to return.

Comparison of Approaches

The tablebased algorithm generated a good set of random numbers. The cycle was
16,384 numbers long. The resulting memsiye was 4 bytes dRAM. For program
memory it required 62 bytes, plus 128 bytes for the initial table, for a total of 190 bytes.

The equatiorbased algorithm also generated a good set of random numbers, and the
cycle was a full 65,536 numbers. The resulting mensing wasalso 4 bytes of RAM.
For program memory it required only 25 bytes. The resulting equation was:

seed =181 *seed + 359 ;

where seed is a 1Bit number. The returnedBit value is the top half of the seed.

" | used the random number generator in Java to populatitie, checking for the EXOR criteria, then
tested the resulting table against all of the grading criteria. Running this code found many acceptable
solutions for the table data; | choose one for my implementation.



Using a Java program on the PC, | searched fgo@d set of constants. The numbers 181
and 359 were used in the equation (both prime numbers), which generated a sequence of
65,536 values that satisfied the grading criteria. Based on the size of the code and the
complexity of the algorithm, | have to obse the equatiebhased algorithm as the best
random number generator code for my applications with the 68HC11 microcontroller.

Testing

The 65,536 sequence of numbers generated by the equation were tested against the
criteria inTablel. | list the first part of this sequence irable3 below. Notice that the
number 172 is repeated, which informs us that a number can be repeated in the sequence.

Table 3. The first part of the 65,536umbers generated by the equation.

1255117 4 73222125232 15167 21110230 252 49 27
35 65133 50218 156 132 185 223 239 99 114 197 223 22 226
226 208 81 76 71229 135182203 4237226 1207119 85
60170 216 82145187 13 4223211228179 190152 202 84 116
50209 28 68 182 28 128 52248 145181 6139210172 63
196 68 28 34183 10119181 56 9 242186 219 229 129 182
243 2216 237 149132106 98 148 78 108 218 134 5210 217
189 13 80 163 78 13 7 89 59 13 94 34102 141 48 159 168
35 99131 69227 27 68 64 161 59 20 94 241 104 232 35
38 6115211 85 56 43113 81228 66 194 176 172172 74
196 244 31 77162226 13206 30 88 171 146 203 251 237 29
254 47135179 20 3 23236 87 6153 81206 66 87 170 156
213181171 5209217199 12 10165 51118 22 190227 199
71136138 67178 39 158 237 43126 81 138 69 50 152 158
85167 38109111 0113250103 34171 11208177201 33

As a follow-on study of the sequence of random numbers, | plotted them as pairs on a
256x256 grid. InFigurel left, three sets of 1024 pairs are plotted as red, blue, and black
points. InFigurel left, half of the 32,76%airs of numbers are plotted. Both of these

plots visually look random, while the right figure is starting to look a bit gife.

L

ba, 0

Figure 1. Three sets of 1024 pairs (left), andlfof the pairs (right) plotted.




In Figure2 below all 32,768 pairs of numbers in the 65,536 set of random numbers are
plotted. This does produce a definite pattern, but it was the least pakenpattern |

found in searching marngairs of candidate constants. Another key point of this set is that
out of 32,768 pairs of numbers, there were 32,768 unique points pldte@ were no
duplicate points from the pairs of numbers.

Figure 2. All 32,768 pairs plotted, resulting in a noticeable pattern.

Most other constants used for the equation covered far fewer positions when plotted, and
generated a much more definite and simple pattern. For example, when using 33 and 171
for the constats, the plot as shown iRigure3 only covers 9728 out of 32,768 positions.

Figure 3. All 32,768 pairs plotted for a different set of constants, resulting in a very noticeable pattern

After extensive testing and looking at both algorithms and different constant values, |
settled on the equatieimased with the constants 181 and 359. This is the algorithm

implemented in the 68HC11 example code.



The Code

Below is 68HC11 assembly code phement the equation to generate random numbers.
All lines starting with **” are comments, and text following a *;’ sign are also comments.
This was assembled with the AS11 assemtded loaded in a 68HC11E1 on a
BOTBoard using PCBUGL11 for testing.

*kkkk *

* Author: Tom Dickens
* Date: 12/10/2002
* Purpose:
*8 - bit random number generator for the 68HC11/68HC12 microcontrollers.
* Method: A 16 - bit version of the classic S=S * M + A is used.
* The random number returned is the high byte of the new seed value.
* The constants M and A were chosen to be M=181 and A=359 for
* the following reasons:
* - They are both prime numbers.
- The equation cycles through all 65,536 possible seed values
- The 8 - bit random values have the following properties:
- Each number from 0 to 255 occurs 256 times in the cycle.
- No sequence of 4 or more numbers is repeated.
- Every 128 - byte window has an average between 104 and 150 (+ - 20%)
- Every 32 - byte window has an average between 76 and 178 (+ - 40)
- Every 8 - byte window has an average between 32 and 218 (+ - 75%)
- Small RAM and Program space required:
- Only 4 bytes of RAM are used
- Only 25 bytes of Program memory is used
- Fast. Each call to Random takes 64 clock cycles, which
is 32 micro - seconds with an 8MHz crystal.
* Usage:
* JSR Random ; Register A contains the next random number.

* * * * * *

EE T R R I R

* * * * * * *

* RAM Usage: 4 bytes in zero - page RAM, bytes 0 trough 3.

* *

ORG $0000
RandomSeed:

RMB 2 ; reserve two bytes in RAM for the SEED value
SEED_HIGH EQU RandomSeed ; high byte of the seed
SEED_LOW EQU RandomSeed+1 ; low byte of the seed
RandomScratch:

RMB 2 ; reserve 2 bytes in RAM for scratch space

* Constants:

MULTIPLIER EQU 181
ADDER EQU 359

* * * * * *

* Program start: Example prog ram to display random numbers on Port C.
* Change ORG $B600 to $F800 for ‘E2 devices.

ORG $B600
LDAA #$FF
STAA $1007 ; Set port C to output

" Version 2.1, dated 38ug-91, by RandySargent.



LDAA  #3$00

STAA $1003 ; Clears port C to $00
BSR Randomlnit ; Seed =0
TopLoop:
BSR Random ; A =random number
STAA $1003 ;A ->PortC
LDX #$FFFF
BSR DelayX ; wait a bit to see the number
BRA TopLoop ; do it again...
*k%k * * *k%k * *k%k *kkkkkkk

* Initialize the random number generator to the start of the sequence
* by loading the seed value with 0. You can initialize the seed to any
* value you choose, which will start the random number sequence somewhere
* else in the cycle of 65,536 se ed values.
Randominit:
CLR SEED_HIGH
CLR SEED_LOW
RTS

Fkk * * * * * *

* Return the next 8 - bit pseudo random number in the se guence
* of 65,536 numbers, generated from the following equation:

* SEED = SEED * 181 + 359

* The random value returned is the high byte of the new SEED.

* Return value: The A register.

* (bytes, cycles) = (25,64)
Random:

PSHB ; (1,3) Remember the current value of B
* scratch = seed * multiplier

LDAA #MULTIPLIER ;(2,2) A=#181

LDAB SEED_LOW ; (2,3) B = the low byte of the seed

MUL ;(1,10)D = AxB

STD RandomScratch ; (1,4) scratch=D

LDAA #MULTIPLIER ;(2,2) A=#181

LDAB SEED_HIGH ; (2,3) B = the high byte of the seed

MUL ;(1,10)D=AxB
* low byte of MUL result is added to the high byte of scratch

ADDB RandomScratch ;(2,3)B =B + scratch_high

STAB RandomScratch ; (2,3) scratch = seed * 181
*

LDD RandomScratch ; (2,4) D = scratch

ADDD #ADDER ;(3,4)D=D + 359

STD RandomSeed ; (2,4) remember new seed value
* (A = SEED_HIGH from ADDD instruction)

PULB ; (1,4) Res tore the value of B

RTS ; (1,5) A holds the new 8 - bit random number
* DelayX: Delay based on the value in.

* * * * * * *%*

DelayX:

PSHX
LoopX:

DEX

BNE LoopX

PULX

RTS




Conclusion

| was able to develop a good random number generator for the 68HC11 that was quite
small (25 bytes) and which met all of my goodness criteria. This was an eqttsau
algorithm using theguation S = S * 181 + 359, with the returned random number the top
8-bits of the 16bit S value. A tablebased algorithm was also developed, which also
generated a good sequence of random numbers, but the algorithm and table required 190
bytes of programace, much larger than the 25 bytes of the equétased algorithm.

The resulting 68HC11 code was successfully tested on an HC11E1 system.

